

# Estructura de decisión

Sitio: [Agencia de Aprendizaje a lo largo de la Vida](#)  
Curso: Tecnicas de Programación 1° F  
Libro: Estructura de decisión

Imprimido por: MARIO DAVID GONZALEZ BENITEZ  
Día: miércoles, 28 de agosto de 2024, 18:50

# Tabla de contenidos


## 1. ¿Qué es una estructura de decisión?

- 1.1. ¿En qué se diferencia una estructura de secuencia de una estructura de decisión?
- 1.2. ¿Cómo se construye una estructura de decisión?
- 1.3. Estructuras de decisión anidadas
- 1.4. Estructura de decisión múltiple

## 1. ¿Qué es una estructura de decisión?




La mayor parte de la potencia del procesador proviene de la capacidad de cálculo, o sea de la capacidad de tomar decisiones y determinar qué acción tomar en el momento de la ejecución del algoritmo. La estructura de decisión consta de realizar una o una serie de acciones entre un conjunto de alternativas.



## 1.1. ¿En qué se diferencia una estructura de secuencia de una estructura de decisión?

### La estructura de secuencia


Sigue una serie de pasos predeterminados. Por ejemplo,



### La estructura de decisión

Debe elegir, a partir de los datos procesados, una alternativa de acción u otra un camino por el cual ejecutar unas sentencias u otras.

Por ejemplo,



## 1.2. ¿Cómo se construye una estructura de decisión?

La construcción comienza con la **palabra SI**, seguida de la **condición** que deba probarse (proposición lógica). La alternativa que debe ser ejecutada si la condición resulta **verdadera** se encuentra precedida por la palabra **ENTONCES**, la otra alternativa (**la falsa**) está precedida por la palabra **SINO**. Esta estructura finaliza con la palabra **FIN SI**.




Como con toda estructura que veamos en el transcurso de la cursada, hacemos hincapié que solo debe tener un único punto de entrada y un único punto de salida. Tiene su entrada donde la condición es evaluada y la salida por fuera de ambas alternativas (centro).




**Veamos un ejemplo práctico:**

- **Situación problemática:** solicitar el ingreso de un número entero que represente la edad de una persona e informar si dicha persona es mayor de edad o no.
- **Hipótesis:** asumimos mayor de edad a partir de los 18 años cumplidos.



Como vemos, se evalúa la condición y este nos arroja un valor verdadero o falso. En base a ese resultado de la condición, el algoritmo tomará un camino u otro.






Veamos otro ejemplo.

Calcular la raíz cuadrada de un número, si este no es negativo, y en caso contrario escribir un mensaje al usuario aclaratorio.

Recordemos que no existe la raíz cuadrada de un número negativo porque ningún número al cuadrado es negativo.




```
Algoritmo raizCuadrada
    Definir num Como Entero
    Definir resultado Como Real
    Escribir "Ingresa un numero: "
    Leer num
    Si num >= 0 Entonces
        resultado = raiz(num)
        Escribir "La raiz cuadrada de ", num, " es ", resultado
    SiNo
        Escribir "No se puede realizar la raiz cuadrada de un numero negativo."
    FinSi
FinAlgoritmo
```

Observemos que en la mayoría de los lenguajes existe una función para obtener la raíz cuadrada de un número que le indiquemos.

En el caso de PSeInt esa función es raíz ([numeroACalcular](#)).

### 1.3. Estructuras de decisión anidadas


Tanto la alternativa verdadera como la falsa, pueden contener a su vez, estructuras de decisión (selección), de secuencia o estructuras de repetición, dentro de ellas. A estas últimas estructuras que están dentro de una alternativa previa de la estructura que las contiene, las llamamos estructuras de decisión anidadas.



Veamos un ejemplo:

Realizar un algoritmo que ingrese dos números enteros y los escriba ordenados de menor a mayor. Si son iguales escribir la leyenda “los 2 números son iguales”.

```
Algoritmo ordenar2Numeros
  Definir num1,num2 Como Entero
  Escribir 'Ingresa un numero: '
  Leer num1
  Escribir 'Ingresa otro numero: '
  Leer num2
  Si num1 == num2 Entonces
    Escribir "Los 2 numeros son iguales"
  SiNo
    Si num1 > num2 Entonces
      Escribir "Ordenados de menor a mayor: ",num2, " - ", num1
    SiNo
      Escribir "Ordenados de menor a mayor: ",num1, " - ", num2
    FinSi
  FinSi
FinAlgoritmo
```






```

Algoritmo diaDeLaSemana
    Definir num Como Entero
    Escribir 'Ingresa un numero correspondiente al dia de la semana: '
    Leer num
    Si num == 1 Entonces
        Escribir "Domingo"
    SiNo
        Si num == 2 Entonces
            Escribir "Lunes"
        SiNo
            Si num == 3 Entonces
                Escribir "Martes"
            SiNo
                Si num == 4 Entonces
                    Escribir "Miercoles"
                SiNo
                    Si num == 5 Entonces
                        Escribir "Jueves"
                    SiNo
                        Si num == 6 Entonces
                            Escribir "Viernes"
                        SiNo
                            Si num == 7 Entonces
                                Escribir "Sabado"
                            SiNo
                                Escribir "Error. Debes ingresar un numero entre 1 y 7"
                            FinSi
                        FinSi
                    FinSi
                FinSi
            FinSi
        FinSi
    FinSi
FinSi
FinSi
FinSi
FinSi
FinSi
FinAlgoritmo

```



Como vemos en el gráfico anterior, resolver esto con la estructura de decisión simple de manera correcta implicaría bastante complejidad y se haría muy confuso. Para evitar este problema surge la estructura de decisión múltiple.



## 1.4. Estructura de decisión múltiple

En esta estructura no existe una [condición Si.. SINO](#); lo que se hace es tomar la comparación de una variable. El valor que pueda tomar esta variable podrá estar o no dentro de las opciones de evaluación de esta [estructura de decisión](#).

La utilizamos cuando se necesita optar entre varias alternativas. Se usa para elegir un camino entre varios alternativos y funciona comparando una variable con una lista de constantes enteras o de carácter. Cuando concuerda con alguna, se ejecuta la secuencia de sentencias asociada con esa constante.



Tomando el ejemplo de los días de la semana, lo representaremos con la estructura de decisión múltiple.

Notamos que el algoritmo es mucho más sencillo que la implementación hecha con la estructura de decisión anidada.

```
Algoritmo diaDeLaSemanaSegun
  Definir num Como Entero
  Escribir 'Ingresa un numero correspondiente al dia de la semana: '
  Leer num
  Segun num Hacer
    1:
      Escribir 'Domingo'
    2:
      Escribir 'Lunes'
    3:
      Escribir 'Martes'
    4:
      Escribir 'Miercoles'
    5:
      Escribir 'Jueves'
    6:
      Escribir 'Viernes'
    7:
      Escribir 'Sabado'
  De Otro Modo:
    Escribir 'Error. Debes ingresar un numero entre 1 y 7'
  Fin Segun
FinAlgoritmo
```

